<commit_before>{ lib, stdenv, fetchurl, which, libao, pkg-config }:

let
  version = "2.13";
in stdenv.mkDerivation {
  pname = "uade123";
  inherit version;
  src = fetchurl {
    url = "http://zakalwe.fi/uade/uade2/uade-${version}.tar.bz2";
    sha256 = "04nn5li7xy4g5ysyjjngmv5d3ibxppkbb86m10vrvadzxdd4w69v";
  };
  nativeBuildInputs = [ pkg-config which ];
  buildInputs = [ libao ];

  enableParallelBuilding = true;
  hardeningDisable = [ "format" ];

  meta = with lib; {
    description = "Plays old Amiga tunes through UAE emulation and cloned m68k-assembler Eagleplayer API";
    homepage = "http://zakalwe.fi/uade/";
    license = licenses.gpl2;
    maintainers = [ ];
    platforms = platforms.unix;
  };
}
<commit_msg>uade123: Add maintainer, 2.13 -> unstable-2021-05-21<commit_after>{ lib
, stdenv
, fetchFromGitLab
, pkg-config
, which
, makeWrapper
, libao
, libbencodetools
, sox
, lame
, flac
, vorbis-tools
}:

stdenv.mkDerivation {
  pname = "uade123";
  version = "unstable-2021-05-21";

  src = fetchFromGitLab {
    owner = "uade-music-player";
    repo = "uade";
    rev = "7169a46e777d19957cd7ff8ac31843203e725ddc";
    sha256 = "1dm7c924fy79y3wkb0qi71m1k6yw1x6j3whw7d0w4ka9hv6za03b";
  };

  postPatch = ''
    patchShebangs .
    substituteInPlace src/frontends/mod2ogg/mod2ogg2.sh.in \
      --replace '-e stat' '-n stat' \
      --replace '/usr/local' "$out"
  '';

  nativeBuildInputs = [
    pkg-config
    which
    makeWrapper
  ];

  buildInputs = [
    libao
    libbencodetools
    sox
    lame
    flac
    vorbis-tools
  ];

  configureFlags = [
    "--bencode-tools-prefix=${libbencodetools}"
  ];

  enableParallelBuilding = true;

  hardeningDisable = [ "format" ];

  postInstall = ''
    wrapProgram $out/bin/mod2ogg2.sh \
      --prefix PATH : $out/bin:${lib.makeBinPath [ sox lame flac vorbis-tools ]}
    # This is an old script, don't break expectations by renaming it
    ln -s $out/bin/mod2ogg2{.sh,}
  '';

  meta = with lib; {
    description = "Plays old Amiga tunes through UAE emulation and cloned m68k-assembler Eagleplayer API";
    homepage = "https://zakalwe.fi/uade/";
    # It's a mix of licenses. "GPL", Public Domain, "LGPL", GPL2+, BSD, LGPL21+ and source code with unknown licenses. E.g.
    # - hippel-coso player is "[not] under any Open Source certified license"
    # - infogrames player is disassembled from Andi Silvas player, unknown license
    # Let's make it easy and flag the whole package as unfree.
    license = licenses.unfree;
    maintainers = with maintainers; [ OPNA2608 ];
    platforms = platforms.unix;
  };
}

{'fraction_non_alphanumeric': 0.1228813559322034, 'fraction_numerical': 0.047380585516178735, 'mean_word_length': 3.141945773524721, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 1, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '3423144', 'n_tokens_mistral': 1046, 'n_tokens_neox': 961, 'n_words': 215}